



UNIVERSITÀ DEGLI STUDI FIRENZE

Scuola di Scienze Matematiche, Fisiche e Naturali  
Corso di Laurea in Matematica

Anno Accademico 2013-2014

22 Luglio 2014

**Localizzazione delle soluzioni di un  
sistema polinomiale zero-dimensionale in  
aritmetica esatta**

**Localization of solutions of a zero-dimensional  
polynomial system in exact arithmetic**

*Candidato:*  
Leonardo Landi

*Relatore:*  
Prof. Giorgio Ottaviani

# Indice

<b>1</b>	<b>Introduzione</b>	<b>2</b>
<b>2</b>	<b>Decomposizione di un'algebra in somma diretta</b>	<b>2</b>
2.1	Ideali zero-dimensionali . . . . .	2
2.2	Decomposizione dell'algebra $\mathcal{A}$ . . . . .	2
2.3	Soluzioni in aritmetica approssimata . . . . .	3
<b>3</b>	<b>Soluzioni reali in aritmetica esatta</b>	<b>4</b>
3.1	La forma traccia . . . . .	4
3.2	Il teorema di Sylvester-Hermite . . . . .	5
<b>4</b>	<b>Polinomi in una variabile</b>	<b>6</b>
4.1	Numero di soluzioni reali in un intervallo . . . . .	7
4.2	Isolamento delle radici reali . . . . .	7
4.3	Calcolo delle molteplicità . . . . .	7
4.4	Approssimazioni per bisezione . . . . .	8
<b>5</b>	<b>Sistemi in due variabili</b>	<b>8</b>
5.1	Numero di soluzioni in un rettangolo . . . . .	9
5.2	Isolamento delle radici reali . . . . .	10
5.3	Approssimazioni per bisezione . . . . .	10
5.4	Calcolo delle molteplicità . . . . .	11
5.4.1	Problematiche nella generalizzazione a più variabili . . . . .	11
5.4.2	Una tecnica di risoluzione . . . . .	11
	<b>Bibliografia</b>	<b>14</b>

# 1. Introduzione

In questo lavoro affrontiamo il problema di studiare le soluzioni reali di un sistema polinomiale zero-dimensionale a coefficienti reali in  $n$  variabili. È possibile in aritmetica esatta localizzare ciascuna soluzione, calcolarne la molteplicità e darne infine un'approssimazione con un livello di precisione arbitrario. È interessante osservare che le tecniche note nel caso di una sola variabile si estendono solo parzialmente al caso di più variabili. Illustriamo alcuni algoritmi che permettono di risolvere il problema costruttivamente per una e due variabili, utilizzando il software *Macaulay2*.

## 2. Decomposizione di un'algebra in somma diretta

### 2.1. Ideali zero-dimensionali

Dato un sistema di equazioni polinomiali  $f_1 = f_2 = \dots = f_m = 0$  nelle variabili  $x_1, \dots, x_n$  a coefficienti in  $\mathbb{C}$ , l'ideale  $I = (f_1, \dots, f_m)$  di  $\mathbb{C}[x_1, \dots, x_n]$  definisce una varietà algebrica affine:

$$\mathcal{V}(I) = \{(a_1 \dots a_n) \in \mathbb{C}^n \mid f(a_1 \dots a_n) = 0, \forall f \in I\}$$

$\mathcal{V}(I)$  rappresenta dunque l'insieme delle soluzioni del sistema. Se queste sono in numero finito, l'ideale  $I$  si dice zero-dimensionale.

**Teorema 2.1.** *Sia  $I$  ideale di  $\mathbb{C}[x_1, \dots, x_n]$ .  $\mathcal{V}(I)$  è finito  $\iff \mathbb{C}[x_1, \dots, x_n]/I$  ha dimensione finita come spazio vettoriale su  $\mathbb{C}$ .*

Per semplicità di notazione, indicheremo con  $\mathcal{A}$  il quoziente  $\mathbb{C}[x_1, \dots, x_n]/I$  e considereremo come base di  $\mathcal{A}$  l'insieme  $\{[x^\alpha] \mid x^\alpha \notin LT(I)\}$  costituito dalle classi dei monomi di  $\mathbb{C}[x_1, \dots, x_n]$  modulo  $I$  (si veda [3, p. 80]). La finitezza di questo insieme è dunque criterio necessario e sufficiente per la zero-dimensionalità di  $I$ .

**Definizione 2.2.** *La dimensione di una base di  $\mathcal{A}$  equivale al numero di soluzioni del sistema definito da  $I$  contate con la loro molteplicità e si chiama grado di  $I$ .*

### 2.2. Decomposizione dell'algebra $\mathcal{A}$

**Definizione 2.3.** *Le matrici  $M_{x_i}$  indotte dalla moltiplicazione per  $x_i$  in  $\mathcal{A}$  si dicono matrici compatte di  $I$ .*

Analogamente, per ogni polinomio  $h(x) \in \mathbb{C}[x_1, \dots, x_n]$  si denota con  $M_{h(x)}$  la matrice indotta dalla moltiplicazione per  $h(x)$  in  $\mathcal{A}$ . Queste applicazioni soddisfano le consuete proprietà di linearità all'interno di un'algebra: rispettano il prodotto per uno scalare e la somma, il prodotto e la composizione di polinomi.

**Definizione 2.4.** Sia  $M$  endomorfismo di  $\mathcal{A}$ . Se  $\lambda_i$  è autovalore di  $M$ , si definisce l'autospazio generalizzato  $\mathcal{A}_i = \cup_{n \in \mathbb{N}} \ker(M - \lambda_i I)^n$ .

La catena  $\ker(M - \lambda_i I) \subseteq \ker(M - \lambda_i I)^2 \subseteq \dots$  per questioni di dimensione diventa stazionaria per  $n$  sufficientemente grande, per cui  $\exists m \in \mathbb{N}$  tale che  $\mathcal{A}_i = \ker(M - \lambda_i I)^m$ .

**Lemma 2.5.** Sia  $M$  endomorfismo di  $\mathcal{A}$ . Allora  $\mathcal{A} = \oplus_{\lambda} \mathcal{A}_{\lambda}$ , dove la somma è estesa a tutti gli autovalori.

**Lemma 2.6.** Dato un insieme finito di punti  $\{p_1, \dots, p_k\} \subset \mathbb{C}^n$ ,  $\exists h(x) \in \mathbb{C}[x_1, \dots, x_n]$  tale che  $h(p_i)$  sono distinti. In tal caso i valori  $h(p_i)$  coincidono con gli autovalori di  $M_{h(x)}$ .

**Teorema 2.7** (Lasker-Noether, 1905). Sia  $\mathcal{V}(I) = \{p_1 \dots p_k\}$  e  $h(x) \in \mathbb{C}[x_1, \dots, x_n]$  che assume valori distinti sui punti di  $\mathcal{V}(I)$ . Siano  $\mathcal{A}_i$  gli autospazi generalizzati di  $M_{h(x)}$ . Allora  $\mathcal{A} = \oplus_{i=1}^k \mathcal{A}_i$ .

Le sottoalgebre  $\mathcal{A}_i$  sono ideali di  $\mathbb{C}[x_1, \dots, x_n]$  e sono provviste di elementi unità  $e_i$ , derivanti dalla decomposizione di 1 in somma diretta, che soddisfano le condizioni di ortogonalità, neutralità e idempotenza.

Il teorema di Lasker-Noether, la cui dimostrazione è ovvia dai due lemmi precedenti, è una riscrittura in altra forma del Teorema di Decomposizione Primaria, dovuto a Emanuel Lasker e provato nella sua generalità da Emmy Noether nel 1921. Per approfondimenti sulla decomposizione primaria e sulle proprietà delle sottoalgebre  $\mathcal{A}_i$  e delle loro unità si veda [2, cap. 4].

**Corollario 2.8.**  $\forall h(x) \in \mathbb{C}[x_1, \dots, x_n]$ ,  $M_{h(x)}$  ristretta ad  $\mathcal{A}_i$  ha come unico autovalore  $h(p_i)$ . Inoltre  $\mathcal{A}_i$  dipende solo da  $I$ .

**Definizione 2.9.** Si definisce molteplicità di  $p_i$  in  $I$  la dimensione di  $\mathcal{A}_i$  e si indica con  $m_{p_i}$ . La molteplicità non dipende dal particolare polinomio  $h(x)$  scelto nel teorema di Lasker-Noether (si veda [1, p. 19]).

### 2.3. Soluzioni in aritmetica approssimata

Prima di enunciare il principale teorema sul calcolo delle soluzioni reali in aritmetica esatta, mostriamo che esiste un metodo per determinare in maniera approssimata le soluzioni di un sistema polinomiale zero-dimensionale mediante lo studio degli autovalori delle matrici compagne.

Già il lemma 2.6 fornisce un criterio per calcolare tutte le  $i$ -esime coordinate delle soluzioni del sistema definito da  $I$ . Infatti queste coincidono con gli autovalori della matrice compagna  $M_{x_i}$ . Tuttavia questo metodo non permette di stabilire quali siano effettivamente le soluzioni del sistema, dal momento che non si può risalire a quali punti appartengano le coordinate trovate. Il seguente teorema pone rimedio a questo problema.

**Teorema 2.10** (Stickelberger). *Sia  $I$  un ideale zero-dimensionale.*

$(\lambda_1, \dots, \lambda_n) \in \mathcal{V}(I) \iff \exists v$  autovettore comune a tutte le matrici compagne  $M_{x_1}, \dots, M_{x_n}$  con autovalori  $\lambda_1, \dots, \lambda_n$  rispettivamente.

Per approfondimenti si veda [1, p. 17,20,21].

## 3. Soluzioni reali in aritmetica esatta

Considereremo d'ora in avanti  $I$  come ideale di  $\mathbb{R}[x_1, \dots, x_n]$  e  $\mathcal{A}$  come il quoziente  $\mathbb{R}[x_1, \dots, x_n]/I$ . Possiamo usare i risultati sulla decomposizione in somma diretta di sottoalgebre trovati per i numeri complessi: in particolare  $\mathcal{A}$  si decompone in sottoalgebre  $\mathcal{A}_i$  che corrispondono a punti reali e in coppie di sottoalgebre  $\mathcal{A}_j, \overline{\mathcal{A}_j}$  che corrispondono a coppie di soluzioni complesse e coniugate.

### 3.1. La forma traccia

Premettiamo un noto risultato sui polinomi in una variabile utile per lo studio della forma traccia:

**Proposizione 3.1** (Regola di Cartesio sui segni). *Se  $f(x)$  è un polinomio a coefficienti reali con tutte radici reali, allora il numero di radici positive (contate con la loro molteplicità) è dato dal numero di variazioni di segno tra due suoi coefficienti consecutivi non nulli.*

Possiamo ora definire la forma traccia come l'applicazione bilineare  $B_h : \mathcal{A} \times \mathcal{A} \rightarrow \mathbb{R}$  tale che  $B_h(a, b) = \text{Tr}(M_{hab})$ .

La sua proprietà di simmetria garantisce che tutti i suoi autovalori siano reali. Pertanto la regola di Cartesio sui segni applicata al polinomio caratteristico di  $B_h$  permette di ricavare il numero dei suoi autovalori positivi. Sempre dal polinomio caratteristico si può ottenere il numero di autovalori nulli, dato dall'esponente del termine di grado minimo e il numero di autovalori negativi per differenza.

La seguente funzione di *Macaulay2* prende in input una qualsiasi matrice simmetrica  $M$  (nella pratica la matrice della forma traccia) e restituisce in output una sequenza di dimensione 3 contenente gli autovalori positivi, negativi e nulli di  $M$ .<sup>1</sup>

```
signature = M -> (
  d := det(M-x*id_(R^D));
  zer := 0; g := d; while gcd(g,x) != 1 do ( zer = zer+1; g = diff(x,g); );
  p := coefficient(x^zer,d);
  pos := 0;
  for i from 1 to D do (
    if coefficient(x^i,d)*p < 0 then ( pos = pos+1; p = coefficient(x^i,d); );
  );
  neg := D - zer - pos;
  return (pos,neg,zer);
);
```

<sup>1</sup>Le variabili globali  $R$  e  $D$  che compaiono all'interno della funzione rappresentano rispettivamente l'anello dei polinomi e la dimensione della base di  $\mathcal{A}$ .

### 3.2. Il teorema di Sylvester-Hermite

Il seguente teorema, enunciato nel caso unidimensionale da Sylvester ed esteso al caso multidimensionale da Hermite, è centrale nel conteggio delle soluzioni di un sistema polinomiale zero-dimensionale e nello studio di alcune caratteristiche come la molteplicità, senza la necessità di ricorrere a metodi numerici.

**Teorema 3.2** (Sylvester-Hermite, 1853). *Sia  $h(x)$  un polinomio di  $\mathbb{R}[x_1, \dots, x_n]$ . Allora:*

1.  $\mathcal{V}(I)$  consiste esclusivamente di punti reali distinti  $p$  tali che  $h(p) > 0 \iff B_h$  è definita positiva.
2. Il numero di punti distinti (reali o complessi)  $p \in \mathcal{V}(I)$  tali che  $h(p) \neq 0$  è uguale al rango di  $B_h$ .
3. La differenza tra il numero di punti reali distinti  $p \in \mathcal{V}(I)$  che soddisfano  $h(p) > 0$  e quelli che soddisfano  $h(p) < 0$  è uguale alla differenza tra autovalori positivi e autovalori negativi di  $B_h$ .

Inoltre, se  $h(p)$  assume valori non nulli sulle coppie di punti complessi coniugati di  $\mathcal{V}(I)$ :

4. Il numero di punti reali distinti  $p \in \mathcal{V}(I)$  tali che  $h(p) > 0$  è uguale al numero di autovalori positivi di  $B_h$  meno il numero di autovalori negativi di  $B_h$ .
5. Il numero di punti reali distinti  $p \in \mathcal{V}(I)$  tali che  $h(p) < 0$  è uguale al numero di autovalori negativi di  $B_h$  meno il numero di autovalori positivi di  $B_h$ .

In particolare che scegliendo  $h = 1$  si può ottenere il numero di soluzioni distinte del sistema e conoscere quante di esse sono reali o complesse.

*Dimostrazione* Per prima cosa osserviamo che la decomposizione di  $\mathcal{A}$  come somma diretta di sottoalgebre è ortogonale rispetto alla forma traccia, nel senso che  $B_h(a, b) = 0 \forall a \in A_i, b \in A_j, i \neq j$ . Quindi la matrice di  $B_h$  nella base di  $\mathcal{A}$  è diagonale a blocchi e si può determinare il suo rango e la sua segnatura come la somma dei ranghi e delle segnature delle restrizioni su ogni sottoalgebra  $\mathcal{A}_i$ .

Possiamo distinguere due casi: quello di sottoalgebra  $\mathcal{A}_i$  corrispondenti a soluzioni reali del sistema e quello invece di coppie di sottoalgebra  $\mathcal{A}_j \oplus \overline{\mathcal{A}_j}$  corrispondenti a soluzioni complesse e coniugate.

**Primo caso:** come base di  $\mathcal{A}_i$  si può scegliere l'insieme

$$\{e_i(x - p_i)^{\alpha_k}, k = 0, \dots, m_{p_i} - 1\}$$

dove  $e_i$  è l'elemento unità di  $\mathcal{A}_i$  e  $\alpha_k \in \mathbb{N}^n$  con  $\alpha_0 = 0$ . Dal Lemma 2.6  $\forall h$  la matrice di  $M_h$  su  $\mathcal{A}_i$  ha come unico autovalore  $h(p_i)$  di molteplicità  $m_{p_i}$ , pertanto la sua traccia è data da  $m_{p_i} h(p_i)$ . La matrice  $B_h$  avrà la forma:

$$m_{p_i} \begin{pmatrix} h(p_i) & 0 & \cdots & 0 \\ 0 & 0 & \cdots & 0 \\ \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & \cdots & 0 \end{pmatrix}$$

pertanto il rango vale 1 se  $h(p_i) \neq 0$  e 0 altrimenti; c'è un unico autovalore positivo se  $h(p_i) > 0$ , negativo se  $h(p_i) < 0$ , nullo se  $h(p_i) = 0$ .

**Secondo caso:** una base per  $\mathcal{A}_j \oplus \overline{\mathcal{A}_j}$  è data da

$$\{e_j(x - p_j)^{\alpha_k} + \overline{e_j(x - p_j)^{\alpha_k}}, -\mathbf{i}(e_j(x - p_j)^{\alpha_k} - \overline{e_j(x - p_j)^{\alpha_k}}), k = 0, \dots, m_{p_j} - 1\}$$

La matrice  $B_h$  assume la forma:

$$m_{p_j} \begin{pmatrix} h(p_j) + h(\overline{p_j}) & -\mathbf{i}(h(p_j) - h(\overline{p_j})) & 0 & \dots & 0 \\ -\mathbf{i}(h(p_j) - h(\overline{p_j})) & -(h(p_j) + h(\overline{p_j})) & 0 & \dots & 0 \\ 0 & 0 & 0 & \dots & 0 \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & 0 & \dots & 0 \end{pmatrix}$$

Il determinante della sottomatrice 2x2 in alto a sinistra vale  $-4 |h(p_j)|^2$  pertanto se  $h(p_j) \neq 0$  il rango vale 2 e c'è un autovalore positivo e uno negativo; altrimenti il rango vale 0 e gli autovalori sono tutti nulli. La differenza tra autovalori positivi e negativi è nulla in ogni caso.  $\square$

Un problema pratico a cui si può trovare risposta grazie al Teorema di Sylvester-Hermite è quello di determinare quante soluzioni ci siano all'interno di un "cubo" dello spazio a  $n$  dimensioni, quale sia la loro molteplicità e se sia possibile approssimarle con precisione arbitraria, ovvero isolarle all'interno di cubetti di lato piccolo a piacimento.

I seguenti algoritmi, implementati in *Macaulay2* nel caso di una e di due variabili danno una risposta a questa domanda. La generalizzazione al caso di  $n$  variabili è immediata dai risultati ottenuti per il caso di due variabili.

## 4. Polinomi in una variabile

Il caso trattato in questo capitolo è quello dello studio delle radici reali di un polinomio  $f(x) \in \mathbb{R}[x]$ . A questo caso si riconducono alcune tecniche di studio delle soluzioni nei casi in cui il numero di variabili è superiore.

Gli algoritmi che seguono utilizzano una funzione `sylvester`<sup>1</sup> creata in precedenza, che prende in input un polinomio, ne calcola la forma traccia e, tramite il calcolo della segnatura con l'algoritmo del capitolo precedente, applica il punto 4) del Teorema di Sylvester-Hermite, restituendo il numero di radici reali  $p$  tali che  $h(p) > 0$ .

---

<sup>1</sup>questa funzione, come tutte quelle elaborate in questo lavoro di tesi, sono consultabili liberamente all'indirizzo <https://github.com/lellolandi/PolySolving/blob/master/PolyServer/calculator.m2>

## 4.1. Numero di soluzioni reali in un intervallo

Sia dato un intervallo aperto  $\mathcal{I}$  della retta reale. Per conoscere il numero di soluzioni interne a tale intervallo è sufficiente applicare la funzione `sylvester` a un adeguato polinomio  $h$  in questo modo:

$$h(x) = \begin{cases} -(x-a) & \text{se } \mathcal{I} = (-\infty, a) \\ x-a & \text{se } \mathcal{I} = (a, +\infty) \\ -(x-a)(x-b) & \text{se } \mathcal{I} = (a, b) \end{cases}$$

## 4.2. Isolamento delle radici reali

Per conoscere la molteplicità delle radici di un dato intervallo e per darne un'approssimazione il punto di partenza è quello di isolare le radici, ovvero di determinare una lista in intervalli interni a  $\mathcal{I}$  disgiunti, ognuno dei quali contenga una sola radice e in modo che ogni radice sia viceversa contenuta in uno di questi intervalli.

La funzione `isolate` svolge questo compito: dati in input i due estremi  $a < b$  (se un estremo è  $-\infty$  o  $+\infty$  si può sempre trovare  $a \in \mathbb{R}$  o  $b \in \mathbb{R}$  tali che l'intervallo  $(a, b)$  contenga le stesse radici poiché queste sono in numero finito) e una lista `intervals` che per i nostri scopi è inizialmente vuota, viene restituita in output la lista di intervalli cercata.

Grazie alla caratteristica di *Macaulay2* di supporto delle funzioni ricorsive, l'algoritmo si presenta nel seguente modo compatto:

```
isolate = (a,b,intervals) -> (  
  if sylvester(-(x-a)*(x-b)) == 0 then return intervals;  
  if sylvester(-(x-a)*(x-b)) == 1 then return append(intervals,{a,b});  
  len := (b-a)/2;  
  while sub(f,x=>a+len) == 0 do len = len/2;  
  intervals = isolate(a,a+len,intervals);  
  intervals = isolate(a+len,b,intervals);  
  return sort intervals;  
);
```

## 4.3. Calcolo delle molteplicità

Sia dato un intervallo  $(a, b)$  contenente una sola radice (ottenuto con l'algoritmo precedente) di cui vogliamo conoscere la molteplicità. L'idea è quella di valutare iterativamente con il Teorema di Sylvester-Hermite quante soluzioni del polinomio  $f_{i+1} = MCD(f_i, f'_i)$ , a partire da  $f_0 = f$ , stanno nel suddetto intervallo. Il primo indice  $k$  per cui  $f_k$  non ha soluzioni in  $(a, b)$  corrisponde alla molteplicità cercata.

Infatti nella fattorizzazione in irriducibili di  $f_{i+1}$  non compaiono i termini relativi alle radici di molteplicità 1, mentre i termini relativi a radici di molteplicità maggiore di 1 compaiono con grado minore di 1 rispetto al grado che avevano in  $f_i$ .



La funzione `multip`<sup>2</sup> prende in input gli estremi  $a$  e  $b$  dell'intervallo e ne restituisce la molteplicità  $m$ .

```

multip = (a,b) -> (
  backupI := I;
  m := 1;
  while gcd(f,diff(x,f)) != 1 do (
    f = gcd(f,diff(x,f));
    I = ideal(f); B = sub(basis (R/I),R); D = numcols B;
    if sylvester(-(x-a)*(x-b)) > 0 then m = m+1
    else break;
  );
  I = backupI; B = sub(basis (R/I),R); D = numcols B;
  return m;
);

```

#### 4.4. Approssimazioni per bisezione

Possiamo dare un'approssimazione delle radici reali di un polinomio facendo uso del Teorema di Sylvester-Hermite, senza ricorrere a metodi numerici integrati nei calcolatori. Ancora una volta il punto di partenza è un intervallo  $(a, b)$  contenente una sola radice. Stabiliamo anche un livello di precisione  $prec$  con cui vogliamo determinare l'approssimazione.

La funzione `bisect` restituisce il valore medio  $m$  di  $(a, b)$  se questo è soluzione (verifica fatta per sostituzione) o se la distanza  $b - a$  è inferiore al doppio del livello di precisione. Altrimenti l'intervallo viene suddiviso in due intervalli  $(a, m)$  e  $(m, b)$  di uguale lunghezza su cui si ripete il procedimento su quello contenente la radice.

```

bisect = (a,b,prec) -> (
  mid := (a+b)/2;
  if (b-a) < 2*prec then return mid;
  if sub(f,x=>mid) == 0 then return mid;
  if sylvester(-(x-a)*(x-mid)) == 0 then return bisect(mid,b,prec)
  else return bisect(a,mid,prec);
);

```

## 5. Sistemi in due variabili

Se da una parte il caso unidimensionale può essere trattato con tecniche che appaiono quasi obbligate, dall'altra risulta difficile e in alcuni casi fallimentare la generalizzazione di queste al caso multidimensionale.


Tuttavia con gli accorgimenti illustrati nel seguito di questo capitolo si riesce a trovare risposta alle medesime problematiche con gli stessi strumenti del Capitolo 3 e senza rinunciare al carattere esatto del risultato.

---


<sup>2</sup>Affinché la funzione `sylvester` operi correttamente, è necessario ridefinire ad ogni passo l'algebra  $\mathcal{A}$ , mediante le variabili globali  $I$ ,  $B$  e  $D$  che rappresentano rispettivamente l'ideale generato da  $f$ , la base di  $\mathcal{A}$  e la sua dimensione. La variabile  $I$  va conservata in memoria all'inizio della procedura per poi ripristinarla al termine.

## 5.1. Numero di soluzioni in un rettangolo


Sia dato un rettangolo  $\mathcal{R}$  del piano delimitato dalle rette  $x = a_x$ ,  $x = b_x$ ,  $y = a_y$ ,  $y = b_y$ . Per determinare il numero di radici interne a tale rettangolo suddividiamo il piano nelle regioni illustrate in figura e denotiamo con i simboli  $\alpha$ ,  $\beta$ ,  $\gamma$ ,  $\delta$  il numero di soluzioni all'interno di ognuna di esse.




$\alpha = \#$  soluzioni nel rettangolo interno



$\beta = \#$  soluzioni nelle strisce verticali esterne




$\gamma = \#$  soluzioni nelle strisce orizzontali esterne




$\delta = \#$  soluzioni nei quattro angoli esterni


Con il Teorema di Sylvester-Hermite, utilizzando appropriati polinomi, possiamo calcolare il numero di soluzioni del sistema presenti nelle zone illustrate in figura:




$n_1$ , ottenuto con  $h = h_1 = -(x - a_x)(x - b_x)$



$n_2$ , ottenuto con  $h = h_2 = -(y - a_y)(y - b_y)$



$n_3$ , ottenuto con  $h = h_1 h_2$



$n_4$ , ottenuto con  $h = -h_1 h_2$

Confrontando le due figure si ottiene il sistema lineare nelle incognite  $\alpha$ ,  $\beta$ ,  $\gamma$ ,  $\delta$ :

$$\begin{cases} \alpha + \beta & = n_1 \\ \alpha + \gamma & = n_2 \\ \alpha + \delta & = n_3 \\ \beta + \gamma & = n_4 \end{cases}$$

Questo sistema, che nel caso generale si presenta con  $2^n$  equazioni in altrettante incognite, con  $n$  numero di variabili del sistema, ha efficacia solo nel caso in cui le rette (o gli iperpiani nel caso di dimensioni superiori) che delimitano la regione non intersechino qualche soluzione del sistema. Nella trascrizione in forma algoritmica di questa procedura abbiamo perciò implementato una funzione `border` che sostituisce i valori  $a_x$ ,  $b_x$  nel generatore (unico) dell'ideale di eliminazione  $I_x$  e i valori  $a_y$ ,  $b_y$  nel generatore (unico) dell'ideale di eliminazione  $I_y$  restituendo un avviso nel caso in cui questi siano soluzioni o procedendo con l'elaborazione del sistema altrimenti. Una volta esercitato questo breve controllo, la funzione `rectangle` restituisce il numero esatto di soluzioni nel rettangolo dato in input (con ovvie notazioni).

```

rectangle = (ax,bx,ay,by) -> (
  nlist := {
    sylvester(-(x-ax)*(x-bx)),
    sylvester(-(y-ay)*(y-by)),
    sylvester((x-ax)*(x-bx)*(y-ay)*(y-by)),
    sylvester(-(x-ax)*(x-bx)*(y-ay)*(y-by))
  };
  S := QQ[a,b,c,d]; use S;
  J := ideal(a+b-nlist#0,a+c-nlist#1,a+d-nlist#2,b+c-nlist#3);
  Jel := eliminate({b,c,d},J);
  use R;
  sol := -sub(Jel_0,a=>0);
  return sol;
);

```

## 5.2. Isolamento delle radici reali

Come nel caso unidimensionale, è necessario isolare le soluzioni in rettangoli interni a  $\mathcal{R}$  che contengano una sola soluzione. L'algoritmo che descrive questo procedimento è analogo a quello illustrato nel caso unidimensionale e si presenta così:

```

subsquare = (ax,bx,ay,by,rects) -> (
  Ix := eliminate(y,I); Iy := eliminate(x,I);
  if rectangle(ax,bx,ay,by) == 0 then return rects;
  if rectangle(ax,bx,ay,by) == 1 then return append(rects,{ax,bx,ay,by});
  lenx := (bx-ax)/2;
  leny := (by-ay)/2;
  while sub(Ix_0,x=>ax+lenx) == 0 do lenx = lenx/2;
  while sub(Iy_0,y=>ay+leny) == 0 do leny = leny/2;
  rects = subsquare(ax,ax+lenx,ay,ay+leny,rects);
  rects = subsquare(ax+lenx,bx,ay,ay+leny,rects);
  rects = subsquare(ax,ax+lenx,ay+leny,by,rects);
  rects = subsquare(ax+lenx,bx,ay+leny,by,rects);
  return sort rects;
);

```

## 5.3. Approssimazioni per bisezione

Fissato un livello di precisione con cui approssimare le soluzioni, la funzione `quadrisect` agisce su un rettangolo dato in input ottenuto in precedenza per isolamento (ovvero contiene una sola radice al suo interno). Ecco la funzione che si occupa di approssimare la soluzione interna a tale rettangolo:

```

quadrisect = (ax,bx,ay,by,prec) -> (
  lenx := (bx-ax)/2; leny := (by-ay)/2;
  midx := ax+lenx; midy := ay+leny;
  Ix := eliminate(y,I); Iy := eliminate(x,I);
  while sub(Ix,x=>midx) == 0 do ( lenx = lenx/2; midx = ax+lenx; );
  if (b-a) >= 2*prec then (
    if rectangle(ax,midx,ay,by) == 0 then return quadrisect(midx,bx,ay,by,prec)
    else return quadrisect(ax,midx,ay,by,prec);
  );
  while sub(Iy,y=>midy) == 0 do ( leny = leny/2; midy = ay+leny; );
  if (by-ay) >= 2*prec then (
    if rectangle(ax,bx,ay,midy) == 0 then return quadrisect(ax,bx,midy,by,prec)
    else return quadrisect(ax,bx,ay,midy,prec);
  );
  return {midx,midy};
);

```

**Nota:** L'analogia col caso unidimensionale è evidente, ma c'è una sostanziale differenza: mentre la bisezione degli intervalli nel caso di polinomi in una variabile è "fortunata" se cade proprio sulla radice di  $f(x)$  (abbiamo infatti trovato la radice in maniera esatta!), nel caso bidimensionale possiamo solo stabilire se la retta che divide il rettangolo in due cade su una coordinata di una soluzione del sistema, ma non se questa coordinata appartiene proprio alla soluzione interna al rettangolo o a una esterna! Per aggirare questo problema è sufficiente eventualmente "spostare" la retta di bisezione, assicurandoci che non accada nuovamente questo fenomeno.

## 5.4. Calcolo delle molteplicità

### 5.4.1 Problematiche nella generalizzazione a più variabili

A differenza degli algoritmi esposti finora, tutti riconducibili ad una generalizzazione del caso unidimensionale, non è possibile estendere con successo l'algoritmo per il calcolo delle molteplicità da una a due variabili. Il procedimento, in linea di principio, apparirebbe così:

1. Parto con un rettangolo  $\mathcal{R}$  contenente una sola radice e definisco l'ideale  $I_0 = I$
2. Itero sull'indice  $i$ , calcolando ad ogni passo  $I_{i+1} = I_i : \sqrt{I_i}$  e controllando con la funzione `rectangle` quante soluzioni di  $I_{i+1}$  stanno in  $\mathcal{R}$ .
3. Il primo indice  $k$  per cui  $I_k$  non ha soluzioni in  $\mathcal{R}$  darebbe la molteplicità cercata

Purtroppo questo metodo fallisce al punto 3), dal momento che la molteplicità di una soluzione per l'ideale  $I_{i+1}$  può abbassarsi di quantità maggiori di 1 rispetto alla molteplicità in  $I_i$ .


**Esempio:** Consideriamo l'ideale  $(x^3, y^2)$  che ha come unica soluzione  $(0, 0)$  di molteplicità 6. Il procedimento descritto sopra si arresta dopo soli 4 passi e non 6, come ci saremmo invece aspettati.

### 5.4.2 Una tecnica di risoluzione

Una tecnica del tutto differente che ha sempre successo si basa sul risultato relativo alle approssimazioni raggiunto nel paragrafo precedente. Illustriamo in sintesi i punti essenziali di questa procedura:

1. Determino un'approssimazione  $\bar{p} = (\bar{x}_0, \bar{y}_0)$  dell'unica soluzione reale  $p = (x_0, y_0)$  presente all'interno del rettangolo di partenza mediante la funzione `quadrisect`, con un livello di precisione  $\epsilon$  stabilito a priori.
2. Chiamo  $A$  e  $B$  gli angoli *NordOvest* e *SudEst* rispettivamente del quadrato di centro  $\bar{p}$  e lato  $2\epsilon$ . La soluzione reale  $p$  sta con certezza all'interno di questo quadrato.

3. Genero un valore random  $a \geq 0$  e considero le due rette del piano appartenenti all'applicazione lineare  $ax - y$  passanti per  $A$  e per  $B$  (si veda la *Figura 1*): queste assumeranno dei valori distinti  $\lambda_A$  e  $\lambda_B$  sui punti  $A$  e  $B$  rispettivamente.
4. Per il Lemma 2.6 gli autovalori della matrice  $M_{ax-y}$  coincidono con i valori assunti dalla forma lineare sulle soluzioni di  $I$ , quindi posso ricondurmi al caso di una variabile, cercando con la funzione `sylvester` quante radici del polinomio caratteristico  $f(t)$  di  $M_{ax-y}$  stanno nell'intervallo  $(\lambda_A, \lambda_B)$ .




**Figura 1** Costruzione del quadrato e scelta della forma lineare


5. Si presentano due casi:

**Caso 1:** Se in  $(\lambda_A, \lambda_B)$  cade una sola radice di  $f(t)$  posso applicare la funzione per il calcolo della molteplicità in una variabile e determinarne la molteplicità, terminando la procedura.

**Caso 2:** Se invece in  $(\lambda_A, \lambda_B)$  cadono due o più radici di  $f(t)$  generiamo un nuovo valore random  $a$  e ripartiamo dal punto 3) (si veda la *Figura 2a*). Può succedere (in casi sfortunati di soluzioni del sistema molto vicine al quadrato, seppure esterne ad esso) che dopo un certo numero di tentativi non si sia riusciti a isolare una sola radice di  $f(t)$  in  $(\lambda_A, \lambda_B)$  pur cambiando il valore di  $a$  (si veda la *Figura 2b*). Occorre quindi ripartire dal punto 1), con un livello di precisione  $\epsilon' < \epsilon$ .



**Figura 2a** La zona di piano tra le due rette intercetta un'altra soluzione  $p'$ . Cambiare il valore  $a$  della forma lineare risulta efficace



**Figura 2b** Cambiando  $a$ , si intercetta sempre qualche soluzione. Occorre quindi ridurre il lato del quadrato e ricominciare

La seguente funzione `multip2` implementata in *Macaulay2* calcola la molteplicità di una soluzione approssimata precedentemente tramite `quadrisect` il cui valore approssimato è `app` con precisione `prec`. Il numero di tentativi di calcolo mediante il cambiamento del parametro  $a$  è impostato a 5. Se non si riesce a trovare un valore

per cui la soluzione compresa tra le due rette è unica, viene restituito il valore 0: si dovrà provvedere a ridurre *prec* e chiamare nuovamente `multip2` con i nuovi valori *app* e *prec*.

```

multip2 = (app,prec) -> (
  backupI := I;
  NW := {app#0-prec,app#1+prec}; SE := {app#0+prec,app#1-prec};
  for i from 0 to 4 do (
    a := random(QQ);
    h := a*x-y;
    v1 := sub(h,{x=>NW#0,y=>NW#1}); v2 := sub(h,{x=>SE#0,y=>SE#1});
    M := companion(h);
    if sylvester(-(h-v1)*(h-v2)) == 1 then (
      R = QQ[x];
      d := det(M-x*id_(R^D));
      I = ideal(d); B = sub(basis (R/I),R); D = numcols B;
      m := 1;
      while gcd(d,diff(x,d)) != 1 do (
        d = gcd(d,diff(x,d));
        I = ideal(d); B = sub(basis (R/I),R); D = numcols B;
        if sylvester(-(x-v1)*(x-v2)) > 0 then m = m+1
        else break;
      );
      R = QQ[x,y]; I = sub(backupI,R); B = sub(basis (R/I),R); D = numcols B;
      return m;
    );
  );
  return 0;
);

```

**Esempio 5.1.** *Studio delle soluzioni reali del sistema:*

$$\begin{cases} 9x^2 - 19xy + 8y^2 + 21x - 12y = 0 \\ 3x^2 - 7xy + 2y^2 + 9x - 6 = 0 \end{cases}$$

L'ideale  $I = (9x^2 - 19xy + 8y^2 + 21x - 12y, 3x^2 - 7xy + 2y^2 + 9x - 6) \subset \mathbb{C}[x, y]$  ha grado 4 e una base per  $\mathbb{C}[x, y]/I$  è  $\{[1], [x], [y], [y^2]\}$ , pertanto la matrice della forma traccia per  $h = 1$  è  $4 \times 4$ . Si osserva che questa ha rango 2, quindi il sistema ammette due sole soluzioni distinte. Il punto 4) del Teorema di Sylvester-Hermite ci dice che queste sono entrambe reali. Potremmo avere una soluzione di molteplicità 1 e una di molteplicità 3 oppure due soluzioni di molteplicità 2. Dallo studio delle matrici compagne ricaviamo informazioni sulle soluzioni: il polinomio caratteristico di  $M_x$  è  $(t-1)^2(t-2)^2$  e quello di  $M_y$  è  $(t-2)^2(t-3)^2$  a meno di coefficienti moltiplicativi. Questo ci dice che ci sono due soluzioni di molteplicità 2, ma non sappiamo quali sono: potrebbero essere (1,2) e (2,3) oppure (1,3) e (2,2). Per esplicitare quale coppia di soluzioni sia quella giusta possiamo ricorrere a più metodi: il più breve è quello di sostituire i valori  $x$  e  $y$  di uno dei punti nei generatori dell'ideale e vedere se questi si annullano. Altrimenti possiamo utilizzare nuovamente il Teorema di Sylvester-Hermite e verificare con il polinomio  $h = (y-x-1)^2$  quante soluzioni stanno nei due semipiani individuati dalla retta  $y = x + 1$ , concludendo che la coppia di soluzioni corretta è (1,2) e (2,3).

# Bibliografia

- [1] G. Ottaviani *Soluzioni di equazioni polinomiali zero dimensionali*. Note reperibili online, versione 30.5.2014
- [2] M. Elkadi, B. Mourrain *Introduction à la résolution des systèmes polynomiaux*. Mathematiques et Applications 59, Springer, Berlin, 2007
- [3] G. Ottaviani *Introduzione alle varietà algebriche Un punto di vista costruttivo*. Disponibile all'indirizzo <http://web.math.unifi.it/users/ottavian/groebner/groebner.pdf>
- [4] D. Cox, J. Little, D. O'Shea *Ideal, Varieties and Algorithms*. Third Edition, Springer, New York, 2007
- [5] D. Cox, J. Little, D. O'Shea *Using Algebraic Geometry*. Second Edition, Springer, New York, 2005
- [6] D. R. Grayson, M. E. Stillman *Macaulay2, a software system for research in algebraic geometry*. Available at <http://www.math.uiuc.edu/Macaulay2>
- [7] D. J. Bates, J. D. Hauenstein, A. J. Sommese, C. W. Wampler *Numerically Solving Polynomial Systems with Bertini*. SIAM, Philadelphia, 2013